

```

1  $PROBLEM PKPD singledose i.v bolus-Inhibitory Indirect response model
2                                     ;DATE 6-2-04 PROGRAMMER:xxxxx
3                                     ;Dose=100mg or 250mg
4                                     ;UNITS: Time=hour, Concentration=ug/ml
5                                     ;kin = ng/ml/hr, kout=hr-1,IC50=ug/ml
6                                     ;Response = biomarker concentration,ng/ml
7
8  $DATA iv1pkpdest_idr1.csv IGNORE=C
9
10 $INPUT ID TIME AMT EFF=DV CMT CLI VI
11
12 $SUBROUTINE ADVAN6 TRANS1 TOL=3
13
14 $MODEL
15     COMP = CENTRAL
16     COMP = EFFECT
17
18 $PK
19     CL      = CLI                ; Individual Clearance in L/hr
20     V       = VI                ; Individual Volume of distribution in L
21     KIN     = THETA(1)*EXP(ETA(1)) ; Basal zero order production rate of biomarker
22     KOUT    = THETA(2)*EXP(ETA(2)) ; Basal first order rate constant for elimination
23                                     ; of biomarker
24     IC50    = THETA(3)*EXP(ETA(3)) ;Concentration of drug at 50% of maximum inhibition
25     S1     = V
26     S2     = 1
27     F2     = KIN/KOUT           ;Baseline response; R0 = kin/kout
28
29
30 $DES
31     DADT(1) = -CL/V*A(1)        ;Plasma compartment
32     INH     = A(1)/(IC50+A(1))  ;Inhibitory function
33     DADT(2) = KIN*(1-INH)-KOUT*A(2) ; Indirect response model I
34
35 $ERROR
36     CP     = A(1)/S1
37     IPRED  = F
38     Y      = F*EXP(ERR(1))
39
40
41 $THETA (1,10)                    ;POPKIN
42 $THETA (0.01,0.5)                ;POPKOUT
43 $THETA (1,20)                    ;POPIC50
44
45 $OMEGA 0.09                      ;BSVKIN
46 $OMEGA 0.09                      ;BSVKOUT
47 $OMEGA 0.09                      ;BSVIC50
48
49 $SIGMA 0.0025                    ;ERRCV
50
51
52 $ESTIMATION METHOD=0 MAXEVAL=9990 PRINT=10 POSTHOC
53 $COVARIANCE
54
55 $TABLE ID TIME DV CP IPRED
56 NOPRINT ONEHEADER FILE=IV1PKPDEST_IDR1.FIT

```

57